

```

1 ' League Fixtures
2 ' by John Holbrow
3 ' Amstrad User Kit (S6)
10 MODE 2:DEFINT a-z
20 INPUT "How many teams in league?";t
30 IF t<2 OR t>100 THEN 10
40 IF t/2<>INT(t/2) THEN t=t+1:by=1
50 DIM fx(t-1,t/2,2)
60 LOCATE 1,3:PRINT"Calculating Week";
70 FOR p=1 TO t-1
80 LOCATE 17,3:PRINT p;
90 x=1:y=t:f=0:w=t-1:k=1
100 FOR c=1 TO w
110 IF c<>p OR x>y THEN 140
120 fx(p,k,1)=x:fx(p,k,2)=y
130 k=k+1
140 IF f=1 THEN y=x:f=0
150 y=y-1:IF y=x THEN y=t:f=1
160 IF y>0 THEN 190
170 IF x=w THEN y=t:f=1:GOTO 190
180 y=t-1
190 NEXT c
200 x=x+1:y=t-x+1:f=0
210 IF x<>t THEN 100
220 NEXT p
230 '
240 ' ***** Display Fixtures *****
250 '
260 LOCATE 1,5
270 PRINT"Output to (S)creen or (P)rinter ?"
280 a$=UPPER$(INKEY$)
290 IF a$="S"THEN s=0:GOTO 320
300 IF a$="P"THEN s=8:GOTO 320
310 GOTO 260
320 PRINT#s:PRINT#s,"Fixtures for";t-by;" teams"
330 FOR x=1 TO w STEP 7: '(step 3 if using 40 cols)
340 k=1:ta=0
350 PRINT#s,:FOR p=x TO x+6: '(x+2 if using 40 cols)
360 IF p>w THEN 420
370 PRINT#s,TAB(ta);fx(p,k,1);
380 IF fx(p,k,2)=t AND by=1 THEN PRINT#s,"= Bye";:GOTO 400
390 PRINT#s,"_";fx(p,k,2);
400 ta=ta+11
410 NEXT p
420 ta=0:k=k+1:IF k<=t/2 THEN 350
430 PRINT#s,:NEXT x
440 PRINT#s
450 PRINT"Another Copy (y/n) ?"
460 a$=UPPER$(INKEY$)
470 IF a$="Y" THEN 270
480 IF a$="N" THEN END
490 GOTO 460

```